Función para el seguimiento de eventos
Muchos eventos ocurren cuando se está operando. Un operador puede ver algunos de ellos directamente en la grafica, por ejemplo, cuando cambia el precio o cuando se cruzan unas líneas de un indicador. Otros eventos, aunque son importantes para el operador, no se ven explícitamente en la grafica. Una gran parte de esos eventos se pueden detectar usando MQL4. 

Por ejemplo, el bróker puede cambiar las condiciones comerciales poco antes de que se publique una noticia fundamental o cuando el mercado está muy activo.  En tales casos, la extensión o la distancia permitida mínima para colocar una orden y para los StopLoss puede aumentarse. Si sucede esto, es necesario, primero detectar las nuevas condiciones y tenerlas en consideración, y en segundo caso, informar al operador sobre estos cambios. 

Para solucionar este problema, usted puede utilizar la siguiente función que maneja este tipo de eventos en su asesor experto o EA. 

Función definida por el usuario Events()
int Events()
Esta función calcula los cambios en la distancia mínima necesaria para crear una orden y para su StopLoss, así como los cambios en la lista (array) de las órdenes abiertas y ordenes pendientes que tiene la cuenta. Para utilizar esta función, se debe utilizar la función Terminal() que vimos anteriormente en otro programa. Esta función utiliza las siguientes arrays globales. 

· Mas_Ord_New - Array que lleva toda la información de las posiciones abiertas y órdenes pendientes. Este array contiene toda la información de cada posición, actualizada desde la última vez de la ejecución de la función Terminal(). 
· Mas_Ord_Old. - Array idéntico al anterior con la salvedad de que va a guardar la información desde la penúltima vez que se ha ejecutado la función Terminal(). 
También se utiliza las siguientes variables globales. 

· Level_new - el valor actual de la distancia mínima;
· Level_old - el valor anterior de la distancia mínima.
Para mostrar los mensajes, la función utilizará a la función de datos Inform(). Si la función Inform() no se incluye en el Asesor de expertos, los mensajes no se verán.
La función que maneja los eventos la vamos a colocar en un archivo externo Events.mqh y luego la incluiremos en nuestro proyecto.  

//--------------------------------------------------------------------------------

// Events.mqh

// Este código solo debe ser usado de manera educacional.
//--------------------------------------------------------------------------- 1 --

// Función para seguir eventos.

// Variables globales:

// Level_new            El ultimo valor de la distancia mínima
// Level_old            El penúltimo valor de la distancia mínima
// Mas_Ord_New[31][9]   Array con el listado de las ultimas ordenes
// Mas_Ord_Old[31][9]   Array con el listado de las penúltimas ordenes
//--------------------------------------------------------------------------- 2 --

int Events()                              // Función definida por el usuario 
  {

   bool Conc_Nom_Ord;                     // Coincidencia de las ordenes  ..

   //.. en las arrays que llevan las ordenes 
//--------------------------------------------------------------------------- 3 --

   Level_new=MarketInfo(Symbol(),MODE_STOPLEVEL );// Ultima distancia conocida
   if (Level_old!=Level_new)              // Si el ultimo no es igual al penúltimo ..

     {                                    // Significa que han cambiado las condiciones 
      Level_old=Level_new;                // Antiguo valor es ahora en nuevo valor
      Inform(10,Level_new);               // Se envía un mensaje con el informe
     }

//--------------------------------------------------------------------------- 4 --

   // Busca ordenes perdidas, cambio den el tipo de orden, partes cerradas y ordenes reabiertas
   for(int old=1;old<=Mas_Ord_Old[0][0];old++)// En el array de las penúltimas ordenes
     {                                    // Suponiendo que..

      Conc_Nom_Ord=false;                 // ..las ordenes no coinciden
      //--------------------------------------------------------------------- 5 --

      for(int new=1;new<=Mas_Ord_New[0][0];new++)//Ciclo en el array de las...

        {                                 //.. ultimas ordenes
         //------------------------------------------------------------------ 6 --

         if (Mas_Ord_Old[old][4]==Mas_Ord_New[new][4])// Si coinciden en el numero
           {                              // El tipo de la orden se convierte en...

            if (Mas_Ord_New[new][6]!=Mas_Ord_Old[old][6])//…diferente
               Inform(7,Mas_Ord_New[new][4]);// Mensaje: modificación:)

            Conc_Nom_Ord=true;            // Las arrays de ordenes coinciden
            break;                        // Se sale del ciclo actual
           }                              

         //------------------------------------------------------------------ 7 --

                                          // El numero de la orden no coincide
         if (Mas_Ord_Old[old][7]>0 &&     // coincide el numero magico
            Mas_Ord_Old[old][7]==Mas_Ord_New[new][7])//.. con el antiguo 
           {               //significa que se a reabierto o se a cerrado parcialmente 
                                             // Si los volúmenes coinciden...

            if (Mas_Ord_Old[old][5]==Mas_Ord_New[new][5])

               Inform(8,Mas_Ord_Old[old][4]);// …es que se ha reabierto
            else                             // En caso contrario..

               Inform(9,Mas_Ord_Old[old][4]);// …se ha cerrado parcialmente 
            Conc_Nom_Ord=true;               // Las arrays de ordenes coinciden...
            break;                           // Se sale del ciclo actual
           }                                 

        }

      //--------------------------------------------------------------------- 8 --

      if (Conc_Nom_Ord==false)               // Si estamos aquí...

        {                                    // ...Esto significa que no encontró orden:(

         if (Mas_Ord_Old[old][6]==0)

            Inform(1, Mas_Ord_Old[old][4]);  // Orden de compra cerrada
         if (Mas_Ord_Old[old][6]==1)

            Inform(2, Mas_Ord_Old[old][4]);  // Orden de venta cerrada
         if (Mas_Ord_Old[old][6]> 1)

            Inform(3, Mas_Ord_Old[old][4]);  // Orden pendiente eliminada
        }

     }

//--------------------------------------------------------------------------- 9 --

   // Buscando en las ultimas ordenes
   for(new=1; new<=Mas_Ord_New[0][0]; new++)// En el array de ultimas ordenes
     {

      if (Mas_Ord_New[new][8]>0)            //Esto no es nuevo pero se ha reabierto
         continue;                          //…o parcialmente cerrado
      Conc_Nom_Ord=false;                   // Las arrays de ordenes no coinciden...
      for(old=1; old<=Mas_Ord_Old[0][0]; old++)// Buscando las ordenes en…
        {                                   // ..el array de las penúltimas ordenes
         if (Mas_Ord_New[new][4]==Mas_Ord_Old[old][4])//Numero coincide..

           {                                          //.. de la orden
            Conc_Nom_Ord=true;              // La orden es encontrada...

            break;                          // Se sale del ciclo actual
           }                                

        }

      if (Conc_Nom_Ord==false)              // Si no se encuentra coincidencia...

        {                                   // ..la orden es nueva:)

         if (Mas_Ord_New[new][6]==0)

            Inform(4, Mas_Ord_New[new][4]); // Orden de compra abierta 
         if (Mas_Ord_New[new][6]==1)

            Inform(5, Mas_Ord_New[new][4]); // Orden de venta abierta 
         if (Mas_Ord_New[new][6]> 1)

            Inform(6, Mas_Ord_New[new][4]); // Orden pendiente colocada
        }

     }

//-------------------------------------------------------------------------- 10 --

   return;

  }

//-------------------------------------------------------------------------- 11 –
Las arrays y variables globales que se utilizaran, son explicadas en el bloque 1 – 2. En el bloque 2 – 3 se declara la variable Conc_Nom_Ord que se usara más adelante para analizar las órdenes. 

La función realiza un seguimiento a los cambios de la distancia mínima para colocar ordenes y StopLoss. Para ello, el valor actual de la distancia mínima Level_new se calcula en cada ejecución de la función (bloque 3 – 4) y, a continuación se compara con el valor anterior que está en Level_old (obtenido durante la anterior ejecución del a función). Si los valores de estas variables no son iguales, significa que la distancia mínima ha sido cambiada por el broker, un poco antes que se ejecutara la función. En este caso, el valor actual de la distancia mínima se asigna a la variable Level_old(se utilizara este valor en la próxima vez que se ejecute la función) y la función Inform() se ejecuta para mostrar el mensaje correspondiente. 
En general, se puede utilizar un método similar a este para detectar otros eventos, por ejemplo, cambios en el spread, el cambio de permisos para negociar una divisa (identificador MODE_TRADEALLOWED en la función MarketInfo()),  el fin de una nueva barra (véase el problema 27 LINK), el cruce de dos líneas de un indicador (véase fig .107 LINK), el llevar a cierta hora, etc. El programa puede detectar algunos eventos utilizando algunos valores del EA para informar al usuario de esto. 
En el bloque 4 – 10 se analiza las órdenes y ordenes pendientes. Se informa al usuario de la mayoría de cambios en una orden. Este análisis se realiza en dos etapas. En la primera fase, el programa detecta cambios como la eliminación y el cierre, cambio de tipo de orden y cerrar cierta parte de una orden o reabrir una orden. En la segunda fase (bloque 9 – 10) se buscan las nuevas órdenes. 

En los bloques 4 – 9 se analizan las órdenes que están en el array Mas_Ord_Old. La cantidad de iteraciones en el ciclo externo “for” depende de la cantidad total de ordenes en el array (elemento Mas_Ord_Old[0][0]) . Para saber si una orden no ha sufrido cambios, es necesario encontrar una orden igual en el array Mas_Ord_New. Esta búsqueda se realiza en el ciclo interno “for” (bloque 6-8), cuya cantidad de iteraciones es igual a la cantidad de ordenes en el array (elemento Mas_Ord_New[0][0]). Más adelante llamaremos a la array Mas_Ord_New “array nueva” y ha Mas_Ord_Old “array vieja”.
En los bloques 6 – 8, el programa busca ordenes, cuyas características sean diferentes. Por ejemplo, en el bloque 6 – 7, se verifica el número de la orden (véase la correspondencia de los índices del array con las características de la orden en la tabla 4 LINK). Si el número de una orden en la array vieja coincide con el número de una orden de la array nueva, significa que por lo menos, esta orden no ha sido cerrada o suprimida. Es necesario también verificar si el tipo de la orden no ha sido cambiado. Si ha sido cambiado, significa que una orden pendiente se ha convertido en una orden abierta.  En este caso, el mensaje correspondiente se a mostrado usando la función Inform(). Independientemente de que se haya cambiado (o se mantenga sin cambios) el tipo de orden, esta orden no será analizada más: el programa sale del ciclo interno y por último, comienza una iteración nueva del ciclo externo. 

Si en el bloque 6 -7 el programa ve que el numero de una orden en la array vieja no coincide con ninguno de los números de ordenes de la array nueva, el programa salta al bloque 7 - 8. Aquí el programa revisa si el número mágico de la orden actual de la array nueva es diferente a cero (todas las ordenes abiertas y pendientes hechas por este EA tienen un numero mágico distinto a cero). Si tiene un número mágico y este coincide con el número mágico de una orden en el array viejo, significa que esta orden esta activa, pero ha cambiado de alguna forma. Hay dos situaciones posibles en que el número de la orden puede haber cambiado. 

Situación 1. Una parte de la orden se ha cerrado. Se puede cerrar una parte de una orden (no pendiente!) en dos etapas según la tecnología aceptada en MT 4. En la primera etapa, la orden inicial es totalmente cerrada. Al mismo tiempo, una orden nueva con un volumen más pequeño se abre al mismo precio y con los mismos precios de StopLoss de la orden inicial. Esta nueva orden tiene un nombre único, distinto al número de la orden inicial. 
Situación 2. La orden ha sido reabierta por el bróker. Algunos bróker (debido a sus normas internas de contabilidad) cierran a la fuerza todas las ordenes al final del día he inmediatamente abren ordenes iguales, pero en el pago corriente y menos intercambio. but at the current price and minus swap. Este evento no afecta para nada los resultados económicos de una cuenta. Cada orden abierta no coincidirá con los números de las órdenes cerradas. 
Las dos situaciones se diferencian en el volumen de las nuevas órdenes: los volúmenes son diferentes en la primera y no cambian en la segunda. Esta diferencia en el volumen se utiliza en el bloque 7 – 8  para distinguir la razón por la que se ha modificado la orden. En ambos casos se mostrara el mensaje correspondiente (“la orden se ha cerrado en parte” o “la orden ha sido reabierta”).  
Si el programa no ha detectado coincidencias (bloque 6 – 7) o parecidos (bloque 7 – 8) en la array nueva, que cierren el ciclo interno, significa que la orden que está en la array vieja ha sido cerrada o eliminada. En este caso, el programa salta al bloque 8 – 9, donde se mostrara un mensaje de acuerdo con el tipo de orden. En el ejemplo anterior, se mostraran tres tipos de mensajes: para una orden de compra, para una orden de venta y para una orden pendiente de cualquier tipo. Si se desea, esta parte puede cambiarse o mejorarse, colocando un mensaje para los diferentes tipos de orden pendientes. 
En la segunda parte del programa (bloque 9 – 10), se analizan las nuevas órdenes en el array nuevo. Esto se hace para detectar órdenes abiertas y pendientes. En el ciclo externo “for”, se revisan las características de todas las órdenes que están en la array nueva. Para identificar las órdenes reabiertas y las cerradas en parte, el programa utiliza una simple característica – la existencia del comentario. Cuando se cierra en parte una orden o se reabre, el servidor del bróker agrega un comentario a la orden donde coloca el número de la orden original. Como en este EA no se agregan comentarios a las órdenes, la existencia de un comentario en una orden indica que la orden no es nueva. 
Si una orden no tiene comentario, el programa busca una orden con el mismo número en la array vieja. Si el programa encuentra un numero de orden que este tanto en la array vieja como en la nueva en el ciclo “for” significa que la orden no es nueva. Pero si encuentra un número en la array nueva que no está en la array vieja, significa que se ha abierto una nueva orden o una nueva orden pendiente. En la parte inferior del bloque 9 – 10 se ejecuta la función Inform() para que muestre el tipo de orden que se ha creado. 

El uso de la función Events() resulta ser muy útil en la práctica. Después que el programador utilice esta función en un EA, se acostumbrara a utilizarla en sus demás proyectos. Debe observase que las funciones Evenys() y Terminal() están estrechamente relacionadas. Si se desea realizar cambios en estas dos funciones (por ejemplo cambiar los nombres de las variables globales), se deberá hacer los cambios en las dos funciones. Si a causa de la estrategia, se necesita colocar comentarios a las órdenes, la parte de este EA que utiliza los comentarios deberá ser adaptada según sea el caso (bloque 9 – 10).

La cantidad de eventos que maneja la función Events() puede aumentarse sustancialmente. Por ejemplo, si se desea mostrar todos los eventos relacionados con las órdenes, se debe analizar las características de las órdenes, como los cambios en el StopLoss de las órdenes abiertas o pendientes o la forma en que se ha cerrado una orden (si la orden se ha cerrado oponiéndola contra otra orden contraria, o se ha cerrado individualmente) y la razón del cierre de una operación o eliminación de una orden pendiente (Si el precio toco un StopLoss, o se ha cerrado manualmente por el operador, etc…). 
